Problem:
Score : 400 points

Problem Statement

We have a large square grid with H rows and W columns.
Iroha is now standing in the top-left cell.
She will repeat going right or down to the adjacent cell, until she reaches the bottom-right cell.

However, she cannot enter the cells in the intersection of the bottom A rows and the leftmost B columns. (That is, there are A×B forbidden cells.) There is no restriction on entering the other cells.

Find the number of ways she can travel to the bottom-right cell.

Since this number can be extremely large, print the number modulo 10^9+7.

Constraints

1 ≦ H, W ≦ 100,000

1 ≦ A < H

1 ≦ B < W

Input

The input is given from Standard Input in the following format:

H W A B

Output

Print the number of ways she can travel to the bottom-right cell, modulo 10^9+7.

Sample Input 1

2 3 1 1

Sample Output 1

2

We have a 2×3 grid, but entering the bottom-left cell is forbidden. The number of ways to travel is two: "Right, Right, Down" and "Right, Down, Right".

Sample Input 2

10 7 3 4

Sample Output 2

3570

There are 12 forbidden cells.

Sample Input 3

100000 100000 99999 99999

Sample Output 3

1

Sample Input 4

100000 100000 44444 55555

Sample Output 4

738162020

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}
import Control.Monad
import Control.Monad.ST
import Data.Array.ST
import Data.Array.Unboxed
import Data.Bits

main = putStrLn . show . solve =<< parse

parse :: IO (Int, Int, Int, Int)
parse = (\[h, w, a, b] -> (h, w, a, b)) . map read . words <$> getLine

solve :: (Int, Int, Int, Int) -> Int
solve (h, w, a, b) = foldl add 0 $ map intrude [b .. w - 1] where
    fact = newFact $ h + w
    inv = newInv fact $ max h w
    ways x y = (fact ! (x + y)) `mul` (inv ! x) `mul` (inv ! y)
    intrude i = (ways (h - a - 1) i) `mul` (ways (a - 1) (w - i - 1))

newFact :: Int -> UArray Int Int
newFact n = runSTUArray $ do
    fact <- newArray_ (0, n) :: ST s (STUArray s Int Int)
    writeArray fact 0 1
    forM_ [1 .. n] $ \i -> do
        n <- readArray fact $ i - 1
        writeArray fact i $ mul n i
    return fact

newInv :: UArray Int Int -> Int -> UArray Int Int
newInv fact n = runSTUArray $ do
    inv <- newArray_ (0, n) :: ST s (STUArray s Int Int)
    writeArray inv 0 1 -- XXX
    writeArray inv 1 1 -- XXX
    writeArray inv n $ pow (fact ! n) (modulo - 2)
    forM_ [n - 1, n - 2 .. 2] $ \i -> do
        x <- readArray inv $ i + 1
        writeArray inv i $ mul x (i + 1)
    return inv

add :: Int -> Int -> Int
add n m = (n + m) `mod` modulo

mul :: Int -> Int -> Int
mul n m = mul' (n `mod` modulo) m 0
mul' n m res
    | m <= 0 = res
    | m .&. 1 == 1 = mul' (shiftL n 1 `mod` modulo) (shiftR m 1) (add res n)
    | otherwise = mul' (shiftL n 1 `mod` modulo) (shiftR m 1) res

pow :: Int -> Int -> Int
pow _ 0 = 1
pow n m
    | m .&. 1 == 0 = mul x x
    | otherwise = mul n . pow n $ m - 1 where
          x = pow n $ m `div` 2

modulo :: Int
modulo = 10 ^ 9 + 7